MorphCol supplement # 27
Calibration for system AMOR2 with JVC camera

18-26 June 2019, corrections 3 September 2019, by Michael Knappertsbusch

Experimental setup:

AMOR2 system, software AMOR v. 4.2 from 27 August 2019 onwards
Leica Mz6
Zoom motor connected to magnification unit of Mz6 by ROTEX GS 5 from KTR
Objective Achromat 1x
Cmount 1x (no lense inserted)
Camera: KY-F75 3CCD digital camera from JVC
Illumination: Polarized light on swan necks, ring-light with poraizer ring, analyzer inserted inside ring-light (cross-polarisation)

Imaging software: NI-MAX version 14.0 from National Instruments
Settings NI Max:
-> Geräte und Schnittstellen -> JVC KY-F75U"cam 0" -> Acquisition attributes:

Video Mode: Format 7, Mode 1, 1360 x 1024
Pixel format: RGB 8 packed
Output image type: AUTO, Speed 400 Mbp
Region of interest: Left 720 Width 640
Top 240 Height 480

Recorded images: 640x480 pixels, 72 dpi, RGB; 1.2MB, tif format.
PC: Optiplex 745 from Dell.
Micrometer scales: The 1mm scale with 0.01mm subdivisions was used (blue box), and the 2mm scale woth 10 micron bars (brown box Hans Schaub) was used as well.

Scales were imaged in horizontal and vertical orientation, at standard physical magnifications of the binocular, i.e. at 0.63x, 0.8x, 1x, 1.25x, 1.6x, 2x, 2.5x, 3.2x and 4x. The pixel distance of scale lengths were then measured using ImageJ, and precision Xprec (in pixel per micrometers) calculated), e.g. distance in pixels divided by distance in micrometers, see Table 1a,b below.

Results

1.) There is a very strong linear relationship for Xprec and Yprec against the magnification (Figure 1). The following two equations are implemeted in program Trace_AMOR3_batch.out

$$\text{Xprec [pixel/µm]} = 0.21292 * \text{Mag} - 0.00037718 \quad r^2 = 0.999 \qquad \text{(I)}$$

$$\text{Yprec [pixel/µm]} = 0.21442 * \text{Mag} - 0.0047747 \quad r^2 = 1.000 \qquad \text{(II)}$$

2.) Xprec and Yprec are almost identical for the JVC camera.



Figure 1: Calibration curves for Xprec and Yprec for system AMOR2 with digital JVC camera.

Data

Mag (X)	Xprec [pixel/micron]	Remarks
0,63	0,132500	1mm scale TF
0,63	0,133250	2mm scale TF
0,63	0,132000	1mm scale MK
0,80	0,165500	1mm scale TF
0,80	0,188500	2mm scale TF
0,80	0,169000	1mm scale MK
1,00	0,211500	1mm scale TF
1,00	0,212000	2mm scale TF
1,00	0,212000	1mm scale MK
1,25	0,267000	1mm scale TF
1,25	0,268750	2mm scale TF
1,25	0,266000	1mm scale MK
1,60	0,335000	1mm scale TF
1,60	0,337780	2mm scale TF
1,60	0,337000	1mm scale MK
2,00	0,424000	1mm scale TF
2,00	0,425710	2mm scale TF
2,00	0,424000	1mm scale MK
2,50	0,536000	1mm scale TF
2,50	0,534780	2mm scale TF
2,50	0,536000	1mm scale MK
3,20	0,666250	1mm scale TF
3,20	0,675625	2mm scale TF
3,20	0,675294	1mm scale MK
4,00	0,856670	1mm scale TF
4,00	0,858333	2mm scale TF
4,00	0,855385	1mm scale MK

Table 1a. Values of Xprec at magnifications 0.63x through 4x from Leica MZ6 binocular, with objective Achromat 1x and JVC camera.
TF=measurement taken by Thore Friesenhagen, MK=measurement taken by Michael Knappertsbusch.

Mag (Y)	Yprec [pixel/micron]	Remarks
0,63	0,133000	1mm scale TF
0,63	0,133000	2mm scale TF
0,80	0,166500	1mm scale TF
0,80	0,167000	2mm scale TF
1,00	0,209000	1mm scale TF
1,00	0,208000	2mm scale TF
1,25	0,263000	1mm scale TF
1,25	0,265290	2mm scale TF
1,60	0,333000	1mm scale TF
1,60	0,335380	2mm scale TF
2,00	0,424500	1mm scale TF
2,00	0,425000	2mm scale TF
2,50	0,536250	1mm scale TF
2,50	0,532500	2mm scale TF
3,20	0,674167	1mm scale TF
3,20	0,678333	2mm scale TF
4,00	0,854000	1mm scale TF
4,00	0,858000	2mm scale TF

Table 1b. Values of Yprec at magnifications 0.63x through 4x from Leica MZ6 binocular, with objective Achromat 1x and JVC camera.
TF=measurement taken by Thore Friesenhagen.

Overview of Trace program versions for AMOR systems:

Trace_AMOR1_batch.out	Trace program for system AMOR with Sony camera; this version has a wrong formula for YPREC.
Trace_AMOR2_batch.out	Trace program for system AMOR with Sony camera with YPREC corrected.
Trace_AMOR3_batch.out	Trace program for system AMOR2 with JVC camera KY-F75.

Program code for Trace_AMOR3_batch.out

```
      PROGRAM TRACE
C
C     !!! Processing in batch mode. !!!
C     From original version 1.2.3, April 28, 1997 by M. Knappertsbusch,
C     modified to batch mode on 21.12.1999
C     modified to write errorneous images to external file. 21.1.2000
C     modified to record missing images, that are read from FILE_LIST. 24.1.2000
C     modified by changing MAG from INTEGER to REAL. 18.12.2000
C     modified by changing the calibration (New XPREC and YPREC). 22.12.2000
C     modified from Trace33_batch.f for application with AMOR using Sony DXC-390P camera.
30.5.2007
C     modified from Trace_AMOR1_batch.f to Trace_AMOR2_batch.f: Correction of wrong YPREC.
13.03.2018
C     modified from Trace_AMOR2_batch.f to Trace_AMOR3_batch.f with system AMOR2, with software
C              AMOR v. 4.2 from 26 April 2019, on Dell Optiplex 745 PC, and
C             using KY-F75 3CCD camera from JVC. 20.06.2019
C     modified from above Trace_AMOR3_batch.f, correction of wrong YPREC. Works now with
C     software AMOR v. 4.2 from 27 August 2019 onwards. 03.09.2019
C
C     Program to trace the outline of a series of digitized light images
C     in batch mode. Input images are grey-level pictures with 480 lines
C     and 680 pixels per line.
C
C
C     !!!!! Output are the X,Y coordinates of the outline
C     in COUNTER-CLOCKWISE direction !!!!!
C     Program includes conversion of pixels in micrometers.
C     Program calibrated to the Leica MZ6 binocular stereo microscope.
C
C     Input files are:
C            - A File with name FILE_LIST containing a list of all
C              names of the input images (filename) to be traced,
C              and the magnifications, that were applied to each image.
C              FILE_LIST is a character variable and can bear any name
C              up to 20 characters in length.
C
C            - All image files to be traced. The names of these raw images
C              are interpreted through the character variable INPUT.
C              INPUT is exactly 5 characters long.
C
C            - The program calls for the sample name through the
C              character variable SAMPLE. SAMPLE is esactly 11 characters
C              long (including orientation codes K or U for Keel or Umbilical).
C
C              INPUT and SAMPLE are used to compose the name of the corresponding
C              output file (suffix _T).
C
C     Output files are:
C            - One file OUTPUT for every traced sample, containing the x,y
C              coordinates of the outline. The character variable OUTPUT is
C              17 characters long (including the suffix _T).
C
C     Note: All file handling for morphometric analyses follows a name convention:
C
C     filename      = Name of sample, 15 characters long.
C     filename_T    = Input file for raw curves with cartesian coordinates, 17 characters long.
C     filename_POL  = Polar coordinates of filename,                        19 characters long.
C     filename_INT  = Interpolated (cartesian) coordinates of filename,     19 characters long.
C     filename_NORM = Normalized curve,                                     20 characters long.
C
C
```

```
C***   Note: MAG must be real, otherwise results may be wrong !
C
       INTEGER I,N,N0,X,Y,THRES,NPTS,NF
       INTEGER GREY(1:8), K(1:8), A(1:8,1:8)
       LOGICAL IMAGEX
       REAL MAG
       DOUBLE PRECISION XPREC,YPREC,XRES,YRES
       CHARACTER*1 CHAR(1:8),ANSWER,COMMA
       CHARACTER*5 INPUT
       CHARACTER*17 OUTPUT
       CHARACTER*11 SAMPLE
       CHARACTER*20 FILE_LIST
       PARAMETER (THRES=49)
C
C      In A(I,J) is the local numbering of points:
C
       DATA ((A(I,J),I=1,8),J=1,8)/1,
     *    2,3,4,5,6,7,8,
     *  2,3,4,5,6,7,8,1,
     *  3,4,5,6,7,8,1,2,
     *  4,5,6,7,8,1,2,3,
     *  5,6,7,8,1,2,3,4,
     *  6,7,8,1,2,3,4,5,
     *  7,8,1,2,3,4,5,6,
     *  8,1,2,3,4,5,6,7/
C
C      Experimental setup and calibration:
C
C       A Leica MZ6 Stereo-Binocular eqipped with a KY-F75 3CCD digital color camera from JVC.
C       System AMOR2, software AMOR v. 4.2 from 27 August 2019 onwards, on Dell Optiplex 745 PC.
C      Zoom motor now connected to microscope with ROTEX GS 5 from KTR.
C       The camera is connected to the microscope via a 1x Cmount (no lense inserted), and
C      Achromat 1x objective is used.
C      Illumination: Polarized light on swan necks, ring-light with poraizer ring, analyzer
C      inserted inside ring-light (for cross-polarisation).
C      Imaging software: NI-MAX from National Instruments.
C      Recorded images: 640x480 pixels, 72 dpi, RGB; 1.2MB, tif format.
C
C      The magnification to be entered is calculated as the product
C      of the objective lens inserted times the number indicated
C      at the zoom-knob. Example: if the objective lens is 1x and the
C      number at the zoom-knob indicates 4.0 then the magnification to be
C      entered is 4.0.
C
C      In the following factors are calculated to convert pixels into µm:
C      (XPREC and XRES are in horizontal, YPREC and YRES are in vertical direction).
C
C      Micrometer scales used for calibration: The 1mm scale with 0.01mm subdivisions was used
(blue box),
C      and the 2mm scale woth 10 micron bars (brown box Hans Schaub) was used as well.
C      Scales were imaged in horizontal and vertical orientation, at standard magnifications of
C      the binocular, i.e. at 0.63x, 0.8x, 1x, 1.25x, 1.6x, 2x, 2.5x, 3.2x and 4x.
C      The pixel distance of scale lengths were then measured using ImageJ, and precision Xprec
C      (in pixel per micrometers) calculated), e.g. distance in pixels divided by distance in
C      micrometers, see Table 1a,b in Knappertsbusch (2019) MorphCol supplement # 27.
C
C      XPREC = 0.21292 * MAG - 0.00037718    (in pixel/µm), equation (I)
C      YPREC = 0.21442 * MAG - 0.0047747     (in pixel/µm), equation (II)
C
C      The necessary resolutions XRES and YRES are the calculated as
C
C      XRES = 1/XPREC     (in µm/pixel)
C      YRES = 1/YPREC     (in µm/pixel)
C
C
```

```
C
      WRITE(9,*) '               Program Trace'
      WRITE(9,*) 'Trace_AMOR3_batch.out, calibration from 18.06.2019'
      WRITE(9,*) ' For system AMOR2, AMOR v. 4.2, JVC camera KY-F75 '
      WRITE(9,*) '            Processing in batch mode'
      WRITE(9,*) ' '
      WRITE(9,*) '       By M. Knappertsbusch, 3 September 2019'
      WRITE(9,*) ' '
      WRITE(9,*) '. . .Enter sample name (11 chars, with K or U). . .'
      READ(9,102) SAMPLE
 102  FORMAT(A11)
C
      WRITE(9,*) '. . .Output in Pixels (1) or Microns (2) ?. . .'
      READ(9,4) ANSWER
 4    FORMAT(A1)
      IF (ANSWER.EQ.'1') THEN
        WRITE(9,*) 'Output is in pixels'
      ELSE
        WRITE(9,*) 'Output is in µm'
        WRITE(9,*) ' '
      END IF
C
C     Get filenames of images and magnifications from FILE_LIST, and start opening
C     and reading for first image:
C
      WRITE(9,*) '. . .Enter list containing image
     1 names (20 chars) and MAG. . .'
      WRITE(9,*) 'Format: 5 chars, comma ,1 real value (2 decimals)'
      WRITE(9,*) 'Note: MAG=Objective lens x zoom factor'
      WRITE(9,*) 'Example: 0101r,3.20 (specimen=0101r,
     2 magnification=3.20x)'
      READ(9,101) FILE_LIST
 101  FORMAT(A20)
C
C
C
C
C     Initialization of NF (NF= number of files traced):
C
      NF=0
C
C     Recording errors during program run:
C
      OPEN(21,FILE='Image_Errors',STATUS='NEW')
      WRITE(21,96) SAMPLE
 96   FORMAT('Errors in Images of Sample: ',A11)
      OPEN(17,FILE='Missing_images',STATUS='NEW')
C
C     Now reading filenames etc:
C
      OPEN(20,FILE=FILE_LIST,STATUS='OLD')
 100  READ(20,110,END=999) INPUT,COMMA,MAG
 110  FORMAT(A5,A1,F4.2)
C
C     Check first, whether image with name INPUT exists or not. If it does not exist:
C     print a message and continue reading next image.
C
      INQUIRE(FILE=INPUT,EXIST=IMAGEX)
      IF (.NOT.IMAGEX) THEN
        WRITE(9,*) 'Error: image ',INPUT,' is missing'
        WRITE(17,97) INPUT
 97     FORMAT(A5)
        WRITE(9,*) '. . .Program continues. . .'
        GOTO 100
      END IF
```

```
C
C       Counting the number of images (=NF) treated:
C
        NF=NF+1
C
C       Initialization of NPTS: (=Number of points in outline):
C
        NPTS=1
C
C       Determination of the names of the OUTPUT files (17 chars long), and
C       writing information to screen:
C
        OUTPUT=(SAMPLE(1:11)//INPUT(1:4))//'_T'
        WRITE(9,111) OUTPUT,MAG
 111    FORMAT('  Specimen: ',A17,' at magnification: ',F4.2,'x')
C
C
C       Determination of XRES and YRES according to MorphCol supplement #27:
C
        XPREC = 0.21292*MAG-0.00037718
        YPREC = 0.21442*MAG-0.0047747
C
        XRES=1/XPREC
        YRES=1/YPREC
C
C
C       WRITE(9,*) '. . .calculating. . .'
C
        OPEN(15,FILE=INPUT,ACCESS='DIRECT',
     *  RECL=1,FORM='FORMATTED',STATUS='OLD')
        OPEN(16,FILE=OUTPUT,STATUS='NEW')
C
C       Find first point of outline in the middle of the
C       TV-screen (this is line 240; its record is calculated
C       as 152960=(240-1)*640):
C
        DO 1, I=1,640
          N=152960+I
          READ(15,50,REC=N) CHAR(1)
          GREY(1)=ICHAR(CHAR(1))
            IF (GREY(1).GE.THRES) THEN
              N0=N
              GOTO 5
            END IF
 1      CONTINUE
C
 5      CALL COORD(N,X,Y)
C       WRITE(9,*) X,Y
C
C       Output of results to file:
C
        IF (ANSWER.EQ.'1') THEN
C         Output of X and Y in pixels:
          WRITE(16,*) X,',',Y
        ELSE
C         Output of X and Y in micrometers:
          WRITE(16,*) X*XRES,',',Y*YRES
        END IF
C
C
C       STOP Criterium if there is an error in the image:
C       (NPTS indicates the number of points written to the outline file; usually,
C       an outline of forams at a magnification of ≤4x has less than 2500 points).
C       The errorneous INPUT images are written into external file with name
C       "Image_Errors", that can be used as input for subsequent runs.
```

```
C
       NPTS=NPTS+1
       IF (NPTS.GT.2500) THEN
         WRITE(9,119) OUTPUT
         WRITE(21,110) INPUT,COMMA,MAG
 119     FORMAT('Error in image ',A17,'. Reading next image')
         GOTO 100
       END IF
C
C
C
C
C
C
C      Local initialization: At each point (pixel) of the outline (record number N)
C      the eight surrounding pixels are read, then their grey-levels
C      are determined, and the pixels are numbered from 1 to 8:
C
C
C             2      3      4
C
C             1      N      5
C
C             8      7      6
C
C
       READ(15,50,REC=N-1)   CHAR(1)
       READ(15,50,REC=N-641) CHAR(2)
       READ(15,50,REC=N-640) CHAR(3)
       READ(15,50,REC=N-639) CHAR(4)
       READ(15,50,REC=N+1)   CHAR(5)
       READ(15,50,REC=N+641) CHAR(6)
       READ(15,50,REC=N+640) CHAR(7)
       READ(15,50,REC=N+639) CHAR(8)
 50    FORMAT(A1)
C
C      Determination of the grey-levels:
C
       DO 10, I=1,8
         GREY(I)=ICHAR(CHAR(I))
 10    CONTINUE
C
C      Assignment of the local pixel number (i.e. 1 through 8)
C      to the record number N for each pixel.
C      (i.e. in K are the record numers of the pixels 1 to 8 stored,
C      with K having the idices from i to 8):
C
       K(1)=N-1
       K(2)=N-641
       K(3)=N-640
       K(4)=N-639
       K(5)=N+1
       K(6)=N+641
       K(7)=N+640
       K(8)=N+639
C
C      Search algoritm for the next point of the outline: The algoritm
C      searches first for the local point to start with: This first
C      point has the grey-level ≥ Threshold. From this initial
C      local point it searches the next point, where the greylevel is
C      lower than Threshold. The pixel, which is previous to that
C      point is the next point of the outline (i.e. the next center for
C      the search algoritm).
C
```

```
C      Look for first point of lokal environment-search:
C
       DO 20, I=1,8
         IF (GREY(I).GE.THRES) THEN
C
C          First point of search path found. Check every next surrounding point in
C          clockwise direction:
C
           DO 30, J=1,8
             IF (GREY(A(I,J)).LT.THRES) THEN
               IF (A(I,J).EQ.1) THEN
                N=K(8)
              ELSE
                N=K(A(I,J)-1)
              END IF
C
C                Stop-Criterium:
C
                IF (N.EQ.N0) THEN
C                 WRITE(9,*) 'Outline closed'
                  CLOSE(16)
                  CLOSE(15)
C
C                 Reading next file:
                  GOTO 100
C
                END IF
C
                GOTO 5
              END IF
 30          CONTINUE
           END IF
 20      CONTINUE
C
 999     WRITE(9,*) ' '
         WRITE(9,*) NF,' images traced and written to files'
         WRITE(9,*) 'One file "Image_Errors" written'
         WRITE(9,*) 'One file "Missing_images" written'
         WRITE(9,*) 'Enter a key to end the program'
         PAUSE 888
 888     CONTINUE
C
         STOP
         END


         SUBROUTINE COORD(M,U,V)
C
C        Determines the cartesian coordinates U and V as a function of
C        the record number M representing a pixel in the image.
C        The image has 480 lines and 640 pixels per line.
C
C        The record number starts with M=1 (representing the first
C        pixel in line 1).
C
         INTEGER M,U,V
C
C        This determination of U and V is valid for all pixels except those
C        that are at the end of each line (i.e. pixel no. 640, 1280,...).
C
         U=MOD(M,640)
         V=479-INT(M/640)
C
         RETURN
         END
```